
Performing DDE Operations with Visual Basic

by Terry Butterfield

Note  Wonderware Technical Support does not support Microsoft® Visual Basic™.  Contact your Microsoft representative for additional information on Visual Basic. 

Microsoft® Visual Basic™ is frequently used to exchange data through DDE.  This Tech Note explains how to perform DDE operations between a Wonderware® InTouch™ application and  a Visual Basic program.  This Tech Note is not a tutorial in Visual Basic and assumes, at the minimum, a fundamental working knowledge of Visual Basic programming.  

This Tech Note will clarify the following Visual Basic concepts and features:

· client and server;

· applications, topics, and items;

· links;

· link attributes;

· events;

· DDE methods and functions.

Note   This Tech Note does not cover implementation of Visual Basic in conjunction with the Wonderware toolkits.

Client and Server  

Two applications can share information in a DDE conversation:  a client and a server.  

· The client is the program that requests the data. In Visual Basic, this program is referred to as the destination.

· The server is the program that contains the data that is being requested.  In Visual Basic, this program is referred to as the source.

It is possible for the same program to both request and provide data.  This program would be a client/server. 

DDE conversations go in both directions between the client and the server.  Depending on the type of link that has been established, if the client changes a value, it will also change the value on the server (even though the server provided the initial value).  

Multiple conversations can be processed at the same time by DDE aware applications.  However, it is important to note that each Visual Basic control establishes its own individual DDE conversation and, as a result, your system performance can be affected by having multiple conversations.  If you need to access a large amount of DDE information, you may want to consider using the Wonderware® Extensibility Toolkit™.

The following Visual Basic controls can be clients:

· text boxes

· picture boxes 

· labels

Any form control can be set up as a server.  The maximum number of controls that can be active on a form at any given time is 128.
Note  Visual Basic picture boxes cannot be used to exchange data with any Wonderware product.

Applications, Topics, and Items

A DDE link requires three pieces of information:  the application name, the topic, and the item.  None of the three are case sensitive.

Application Name

The application name is the name of the executable file, such as View.  Do not include the file extension or the path to the file.  This is because DDE is dealing with an instance of the file in memory.  Entering a file extension or path will cause an error to occur.  

If you are trying to create a DDE link with another node on the network through NetDDE, then the application name should be entered as follows: 

\\nodename\application
where nodename is the name of the computer on the network and application is the name of the application you are trying to access.  For example, \\B2South\Excel would attempt to access a computer that has a node name of  B2South and then, once the computer was found, it would try to access the instance of Microsoft® Excel™ that is running on that machine.

Topic

The topic specifies the subject of the conversation.  For example, if Wonderware® WindowViewer™ was the subject of the conversation, then the topic would be the word Tagname.  Or, if a Wonderware I/O Server was the subject of the conversation, the topic would be one of the topics that are defined under the Topic Definition menu.  If you are building a Visual Basic source application, each form has its own topic name (See the description of LinkTopic on the next page).  

Item
The item specifies the piece of data that is to be exchanged.  In InTouch, these can be the tagnames you create.  In a Visual Basic source form, this can be the name of any DDE aware control.

Note  To connect to a Visual Basic program during design time, make sure the file name that is given is the same as the project name.  

Links     

Frequently the DDE conversation itself is referred to as a link.  This is because the client is linked to the server by the data they are sharing.  Visual Basic recognizes three types of links:

· a hot link, also known as an automatic link, where the data is updated  whenever it changes on either the client or the server (this is link mode 1);

· a warm link, also known as a notify link, where the client program is notified when the DDE data has changed on the server.  It is then up to the client program to request the data from the server program (this is link mode 3);

· a cold link, also known as a manual link, where data is updated only upon request (this is link mode 2).

Note  Wonderware does not support the warm link method; therefore, any attempt to use it may produce undefined results.

Link Attributes

Each control that is placed on the Visual Basic form (and also the form itself) has various attributes.  The attributes that we are concerned with are:

· LinkTopic

· LinkItem

· LinkMode

· LinkTimeout

Here are the attributes for a text box before they are defined (using Visual Basic, version 4):  




How these attributes are defined will determine how the Visual Basic application will handle DDE.

LinkTopic

If the Visual Basic program is being used as a client for DDE, the LinkTopic is used to specify the name of the source application and the topic of the conversation.  The syntax for this LinkTopic would be app|topic, where app is the name of the application that is being linked to, such as View, and topic is a valid topic in that application.  In the case of InTouch, the syntax for this line would be View|Tagname.  The LinkTopic is the unique name of the link and cannot be changed at runtime without breaking the link.  

Example:
LinkTopic
View|Tagname

If the Visual Basic program is being used as a server for DDE on a form object, the LinkTopic can be named anything, or by default, it will be the name of the form.  From Excel you can now access the Visual Basic application as follows:

 =app|topic!controlname

app is the name of the Visual Basic application, topic is the LinkTopic property of the form you want to access (the LinkTopic for that form must be defined), and controlname is the name of a DDE aware control on that form.  

Under Visual Basic, versions 3, 4 and 5, the standard DDE aware controls are:  text box, picture box and  label.

Example:
LinkTopic
VB_Topic

LinkItem

This defines which data is to be exchanged.  The data that is entered on this line could be one of the following:  a valid tagname in the InTouch tagname dictionary, a valid item name for a DDE server, or a valid item in any other DDE aware program.  You can change the LinkItem at runtime without breaking the link.

Example:
LinkItem
$Second

LinkMode

This defines the type of DDE link that is to be established (hot, warm or cold).  The link mode can be set to 0, which disables DDE for that control, or it can be set to the number for the type of link you want to establish. 

Note  If LinkMode for a server Visual Basic form is set to 0 at design time, it cannot be changed at run time.  

If  the LinkMode is set to a non-zero value, at run time it can be changed to 0 and then back to another valid link number. For a client application, this value does not need to be set at design time.

Example:
LinkMode
2

LinkTimeout

This specifies the amount of time, in tenths of a second, that Visual Basic will wait for a response from the source before Visual Basic will generate an error message.  The default LinkTimeout is five seconds, shown in the example below as 50 tenths of a second.

Example:
LinkTimeout
50

The Sequence for Setting Link Attributes

When using Visual Basic code to set up the controls, the link attributes should be set in the following order:

1. Set the LinkMode to 0

2. Set the LinkTopic

3. Set the LinkItem

4. Set the LinkTimeout (this step is optional)

5. Set the LinkMode to 1, 2 or 3 

After setting these attributes, the client application will attempt to make a DDE link to the server application.  The LinkMode should always be set to 0 before setting it to any other value.  Not setting the properties in sequence may prevent a control(s) from establishing a DDE conversation.

Events

As with any other Visual Basic control, there are events that allow you to control the program flow in your application.  The events we are concerned with include the following:

· LinkOpen

· LinkClose

· LinkError

· LinkNotify

LinkOpen

When a control successfully initiates a link to a client application, the client application receives a LinkOpen event.  Similarly, when a client application successfully initiates a link to a server application, the server application receives a LinkOpen event.  The server application can accept or cancel the LinkOpen event.  

The LinkOpen event for a server application applies to form events only.  It is possible to use a control on the server form as a client control.  In this case, the LinkOpen event would apply to both of the controls independently.

LinkClose

A LinkClose event applies to both client and server forms and controls.  The LinkClose event will occur if either the client or server closes the link.  

Note  For proper termination of a DDE conversation by the Visual Basic client or server application, the LinkMode should be set to 0.
LinkError

If Visual Basic is used to manipulate a DDE conversation, errors can become trapped and get processed just like any other run-time error.  For example, a client application will receive a run-time error when it attempts to link to a server application that is not available or if the topic is not valid (For more information on errors and their meanings, do a search on “run-time errors,” “trappable errors,” and “DDE and Form-Related errors”  in the Visual Basic on-line help.).

LinkNotify 

Since Wonderware products do not support notify links, or warm links, there is no use for this attribute when dealing with a Wonderware application.

Methods and Functions

Methods and functions can be used to manipulate other applications through a DDE link.  Once a link is established, you can send data, request data, and send the server application commands.  When a Visual Basic application is set up as a server, it can notify the client when the data has changed.  The methods and functions were are concerned with include the following:

· LinkPoke

· LinkExecute

· DoEvents
LinkPoke

A LinkPoke creates a link to the client application and writes data to it.  Data will be written to the client application once per LinkPoke.  

LinkExecute

A LinkExecute can send commands to programs that support DDE.  For example, a LinkExecute can be used to trigger macros in Excel.  However, Wonderware products do not respond to this command.

DoEvents 

The DoEvents function will cause Visual Basic to yield execution so that the operating system can process other events.  When a Visual Basic application is waiting for an event to occur, the system resources are automatically shared.  Visual Basic also yields resources when it performs a DDE operation.  Most applications respond at this time.  Some applications may be unable to act on their side of the DDE conversation.  This will cause a run-time error (time-out) in the Visual Basic application.  To help avoid this situation, call the DoEvents function.  The DoEvents function should be used under the following conditions:

· When you are running processor-intensive code.  This will yield control to the Windows operating system at the time that the DoEvents statement is made.

· When you want to give control to a different process for a while, such as when you issue a LinkExecute Method to Excel.

Sample Visual Basic Application

The following steps describe how to build a Visual Basic application that will interface with a simple InTouch application.  Note that in these steps, the Properties dialog boxes have been edited to show only the attributes that were changed in each step. 

Step 1

Open a new Visual Basic application and then enter the properties for the new form as shown in the Properties dialog box to the right of the form.  




· The text that is entered for Caption will determine the text that is displayed on the title bar.  

· Setting LinkMode to 1-Source will allow the form to respond to client applications that are requesting data.  

· Setting LinkTopic to VB1 sets the topic for this form to Visual Basic1.  

· Setting Name to frmDDESam changes the name of the form.  When naming a control (in this case, it is the form) it is generally a good programming practice to put a prefix in front of the name that denotes what type of control is being named.  For example, frm indicates that we are naming a form.

Code for the frmDDESam:  

For the event LinkClose, insert the following:

txtModified.BackColor = &HFF&



‘Set Background to red when server link closes

For the event LinkOpen, insert the following:

txtModified.BackColor = &H8000&


‘Set Background to green when link is opened

For the event Load, insert the following:

frmDDESam.LinkMode = 1



‘Turn on the form as a server

txtModified.Tag = 0

Step 2

Add a text box to the form.  This text box will be used to look at the InTouch system tag $Second through a hot DDE link. Change the properties on the text box as shown in the Properties dialog box to the right of the form.  Entering these properties will make the Visual Basic control become a client that requests data from the server, which will be InTouch in this example.




The LinkTopic shows the server name and topic name that the client will communicate with.  In this example, the application is View and the topic is Tagname.  The LinkItem is the item name or data location that is to be read.  In this case, the InTouch system tag $Second.is to be read.  Note that these attributes must be filled in before setting the LinkMode.  

After the LinkMode is set to a non-zero value, Visual Basic will attempt to establish a DDE connection, even during design time, to the specified links.  The program you are attempting to link to must also be up and running at this time.  (This allows the link to be established).  For a client application, these attributes can be set at run time.  If the application is going to act as a server, the LinkMode property that is on the form that contains the data you want to share must be set to 1 at design time.
Code for txtSecond:

For the event Change, insert the following:

txtModified.Text = txtSecond * 1.32  


‘Modify the value you read in.

For the event LinkClose, insert the following:

txtSecond.BackColor = &HFF&



‘Set Background to red when server link closes

For the event LinkOpen, insert the following:

txtSecond.BackColor = &H8000&



‘Set Background to green when link is opened

Step 3 

Add a second text box to the form.  This text box will be used to provide information to InTouch. Change the properties on the text box as shown in the Properties dialog box to the right of the form. This text box will cause the Visual Basic program to act as a sever as well as a client.




This text box is going to provide InTouch an item to read.  The ItemName will be the name of the control.  In this case, the control is txtModified.  This is going to allow the sample application to act as a server that is responding to the clients request.

Code for txtModified:  There is no code for txtModified.
Step 4 

Add a command button to the form, using the properties that are shown in the Properties dialog box.  This button is going to be used to open the communication between txtSecond and the InTouch tag $Second.




Code for cmdClient:
For the event Click, insert the following:

If txtSecond.BackColor = &HFF& Then

  txtSecond.LinkMode = 0                



'Setups

  txtSecond.LinkTopic = "View|Tagname" 


'Application|Topic

  txtSecond.LinkItem = "$Second"       



'Item

  txtSecond.LinkTimeout = 10           



'Timeout

  If optLinkType(1).Value = True Then

    txtSecond.LinkMode = 1             



'Set mode to automatic

  Else

    txtSecond.LinkMode = 2              



'Set mode to manual

  End If

Else

 txtSecond.LinkMode = 0                 



'Break Link

End If

Step 5 

Add a command button to the form using the properties shown in the Properties dialog box.  This button is going to be used to update data between txtSecond and the InTouch tag $Second when a cold link is established.




Code for cmdRequest:
For the event Click, insert the following:

On Error GoTo Reqerror

  txtSecond.LinkMode = 0

  txtSecond.LinkTopic = "View|Tagname"

  txtSecond.LinkItem = "txtSecond"

  txtSecond.LinkMode = 2

  txtSecond.LinkRequest

  txtSecond.LinkMode = 0

Reqerror:

Select Case Err

  Case 282

    MsgBox "Foreign application " & txtModified.LinkTopic & " did not respond to a DDE initiate"

  Case 286

    MsgBox "Timeout waiting for a response."

  Case 292

    MsgBox "DDE Conversation Closed or changed."

  Case 293

    MsgBox "DDE Method attempted without DDE channel open."

  Case Else

End Select

Resume Next

Step 6

Add a command button to the form using the properties shown in the Properties dialog box.  This button is going to be used to open and close access to the form as a server.  Note that while the form is closed as a server, the client portion of the program, txtSecond, will still be able to act as a client.




Code for cmdServer:

For the event Click, insert the following:

If txtModified.Tag = 0 Then

  frmDDESam.LinkMode = 1                

'Make Server available

  txtModified.Tag = 1

Else

  frmDDESam.LinkMode = 0                 

'Break Link

  txtModified.Tag = 0

End If

Step 7 

Add a command button to the form using the properties shown in the Properties dialog box.  This button is going to be used to poke data into InTouch.  When the form is no longer acting as a server with a hot link, data can still be sent to the server with a cold link.




Code for cmdPoke:

For the event Click, insert the following:

On Error GoTo PokeError

  txtModified.LinkMode = 0

  txtModified.LinkTopic = "View|Tagname"

  txtModified.LinkMode = 2

  txtModified.LinkPoke

  txtModified.LinkMode = 0

PokeError:

Select Case Err

  Case 282

    MsgBox "Foreign application " & txtModified.LinkTopic & " did not respond to a DDE initiate"

  Case 286

    MsgBox "Timeout waiting for a response."

  Case 292

    MsgBox "DDE Conversation Closed or changed."

  Case 293

    MsgBox "DDE Method attempted without DDE channel open."

  Case 297

    MsgBox "Cant set LinkMode; Invalid link topic."

  Case Else

End Select

Resume Next

Step 8 

Add an option button to the form using the properties shown in the Properties dialog box.  This option button will be used to switch between Manual links and Automatic links.




Code for optLinkType(0):  There is no code for optLinkType(0).
Step 9 

Add an option button to the form using the properties in the Properties dialog box.  This option button will be used to switch between Manual links and Automatic links.  This defines the form and the controls.




Code for optLinkType(1):  There is no code for optLinkType(1).
This completes the visual basic application.  

InTouch Application for Use With Visual Basic

The InTouch application that you should use to talk to the Visual Basic application is as follows:

Where:

· # is a value display analog, DDE integer, with an Item Name of txtSecond;

· #####.##### is a value display analog, DDE real, with an Item Name of txtModified.

Set up the DDE Access name as follows.  Use this DDE Access name for both DDE tags.

Depending on the version of Visual Basic that you are working with, it may be easiest to link to the compiled version of the Visual Basic application.

When the Visual Basic and InTouch applications are run, the Visual Basic application will read the tag $Second from the InTouch Application and then display this value in txtSecond.  Next, the Visual Basic application will display the results of txtSecond*1.32 in txtModified.  By clicking on Open/Close Client, you can cause the Visual Basic application to stop requesting $Second.  By clicking on Open/Close Server, you can prevent the Visual Basic application from providing data to the InTouch application.  Notice that after you close the Server, conversation is not automatically restarted when you open the server.  You must do a ReInitialize DDE from the InTouch application since it is the client that requests the data.

If the txtSecond text box is red, no DDE conversation is established.  If the txtModified text box is red, no application is attempting to talk to the Visual Basic application, or the form is closed as a server.

Additional Information

See the following references for additional information on Visual Basic programming:

· If you are using Visual Basic, Version 3, see Chapter 21 of the Microsoft Visual Basic Programmers Guide.

· If you are using Visual Basic, Version 4, see Chapter 25 of the Microsoft Visual Basic Programmers Guide.

· If you are using Visual Basic, Version 5, refer to Visual Basic Books Online and search for DDE.

For additional information on the Internet, see the following web sites:

Microsoft
-- http://www.microsoft.com/kb/

(Type the words Visual Basic and DDE in the search box, then click on the Search button.)

Visual Basic Programmers Journal --
http://www.windx.com/scripts/homepage.epl/
  
(Type DDE in the Site Search box, then click on GO!)

